




## Experiment #23: Expanding Outputs

This experiment demonstrates the expansion of BASIC Stamp outputs with a simple shift register. Three lines are used to control eight LEDs with a 74x595 shift register.

### New PBASIC elements/commands to know:

- SHIFTOUT

**Building The Circuit** (Note that schematic is NOT chip-centric)



## Experiment #23b: Expanded Outputs

---

```
' =====
'
' File..... Ex23 - 74HC595.BS2
' Purpose... Expanded outputs with 74HC595
' Author.... Parallax
' E-mail.... stamptech@parallaxinc.com
' Started...
' Updated... 01 MAY 2002
'
'   {$STAMP BS2}
'
' =====
'
' -----
' Program Description
' -----
'
' This program demonstrates a simple method of turning three Stamp lines into
' eight outputs with a 74HC595 shift register.
'
' -----
' I/O Definitions
' -----
'
Clock          CON      0          ' shift clock (74HC595.11)
DataOut        CON      1          ' serial data out (74HC595.14)
Latch          CON      2          ' output latch (74HC595.12)
'
' -----
' Constants
' -----
'
DelayTime      CON      100
'
' -----
' Variables
' -----
'
pattern        VAR      Byte      ' output pattern
```

## Experiment #23: Expanded Outputs

```
'-----  
' Initialization  
'-----  
  
Initialize:  
  LOW Latch                                ' make output and keep low  
  pattern = %00000001  
  
'-----  
' Program Code  
'-----  
  
Go_Forward:  
  GOSUB Out_595  
  PAUSE DelayTime                          ' put pattern on 74x595  
  pattern = pattern << 1                    ' shift pattern to the left  
  IF (pattern = %10000000) THEN Go_Reverse ' test for final position  
  GOTO Go_Forward                          ' continue in this direction  
  
Go_Reverse:  
  GOSUB Out_595  
  PAUSE DelayTime  
  pattern = pattern >> 1  
  IF (pattern = %00000001) THEN Go_Forward  
  GOTO Go_Reverse  
  
'-----  
' Subroutines  
'-----  
  
Out_595:  
  SHIFTOUT DataOut, Clock, MSBFirst, [pattern] ' send pattern to 74x595  
  PULSOUT Latch, 5                             ' latch outputs  
  RETURN
```

### Behind The Scenes

The BASIC Stamp is extraordinarily flexible in its ability to redefine the direction (input or output) of its I/O pins, yet very few applications require this flexibility. For the most part, microcontroller applications will define pins as either inputs or outputs at initialization and the definitions will remain unchanged through the program.

## **Experiment #23b: Expanded Outputs**

---

We can use the fact that outputs are outputs and conserve valuable BASIC Stamp I/O lines at the same time by using a simple component called a serial-in, parallel-out shift register. In this experiment, the 74x595 is used. With just three BASIC Stamp lines, this program is able to control eight LEDs through the 74x595.

The 74x595 converts a synchronous serial data stream to eight parallel outputs. Synchronous serial data actually has two components: the serial data and a serial clock. The BASIC Stamp's **SHIFTOUT** command handles the details of the data and clock lines and writes data to a synchronous device, in this case, the 74x595. With the 74x595, the data must be latched to the outputs after the shift process. Latching is accomplished by briefly pulsing the Latch control line. This prevents the outputs from "rippling" as new data is being shifted in.

Being serial devices, shift registers can be cascaded. By cascading, the BASIC Stamp is able to control dozens of 74x595 outputs with the same three control lines. To connect cascaded 74x595s, the clock and latch lines are all tied together and the SQ output from one stage connects to the serial input of the next stage.